
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 12.4.2.3b which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_12_4_2_3b
Test Group:
/12/UTRAN34_GMM_SIMTC.ttcn

ATS Version:
iwd-TTCN3-B2013-03_D16wk24 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Intel 7360
Verification Status:
PASS
4 Corrections required for test case 12.4.2.3b
4.1 Introduction

This section describes the changes required to make test case 12.4.2.3b run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D16wk24 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 f_TC_12_4_2_3b_UTRAN()
	Record Name
	f_TC_12_4_2_3b_UTRAN()

	Reason for change
	In the current TTCN implementation of this function, the UE is simply switched ON as per Step 3 of expected sequence in 34.123-1 v 12.4.0 Sec 12.4.2.3b.4. This could potentially mean that the UE may have a P-TMSI, or a TMSI because of last operation. The initial conditions for the testcase (34.123-1 v 12.4.0 Sec 12.4.2.3b.4) clearly state that UE should have a valid IMSI. The current TTCN does not ensure this.

	Summary of change
	The functions f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell),f_UTRAN34_AttachReject(utran34_CellA, IMSI),f_UTRAN34_SwitchPower_Off(utran34_CellA, U1_IDLE, IMSI_DetachOnly) are called in this specific order to ensure that the UE performs an ATTACH REJ and only has an IMSI as UE identity. 

Please see screenshot for details.

	Source of change
	UTRAN34_GMM_SIMTC.ttcn

	
	


Before:
	<<SKIPPED CODE>>

//Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellB);
    //Configure UE to use SMS-only service
    f_UT_MMI_ConfigureSMSOnly(UT);

<<SKIPPED CODE>>


After:

	<<SKIPPED CODE>>

//Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellB);
    //WA#12_4_2_3b : Preamble condition to ensure that UE has a IMSI    
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);
    f_UTRAN34_AttachReject(utran34_CellA, IMSI);
    f_UTRAN34_SwitchPower_Off(utran34_CellA, U1_IDLE, IMSI_DetachOnly);
    //WA#12_4_2_3b : Preamble over
    //Configure UE to use SMS-only service
    f_UT_MMI_ConfigureSMSOnly(UT);

<<SKIPPED CODE>>


4.3 fl_TC_12_4_2_3b_Body()
	Record Name
	fl_TC_12_4_2_3b_Body()

	Reason for change
	1) In the current TTCN implementation, the ATTACH REQ expected as per Step 4 of the test sequence in 34.123-1 v12.4.0 Sec 12.4.2.3b.4, expects RAI-1. Since the UE has performed ATTACH REJ in preamble, the UEwont be sending the RAI-1 information. This needs to be corrected in current TTCN.
2) The current TTCN implementation of ATTACH ACCEPT message (Step 8 of expected sequence as per 34.123-1 v12.4.0 Sec 12.4.2.3b.4) transmits the IE GMM cause. As per 24.008 Sec 9.4.2 the IE GMM Cause is a TV type. However, the current implementation does not include the IEI for this. This needds to be corrected.

3) The current TTCN implementation does not release the rrc connection before changing the power levels as per Step 10 of the test sequence in 34.123-1 v12.4.0 Sec 12.4.2.3b.4. The TTCN straightaway reconfigures both the cells for the behaviour  expected as per Steps 12  to 14 of  the test sequence. This is incorrect and can cause a RL failure for UE. This needs to be corrected.

4) The current TTCN expects the RAU REQ at Step 12 as an UplinkDirectTransfer message. However, because of change mentioned in point 3 above, the RAU REQ will come as an InitialDirectTransfer, not UplinkDirectTransfer. This needs to be corrected. Also, the RAU REQ expects the Update type as “RA Updating”. This is incorrect since the cells are in NMO I. The update type needs to be changed to “Combined RA/LA with IMSI attach”. This is in line with 24.008 Sec 4.7.5.2.1. A Prose CR will be raised for this.
5) The current TTCN does not implement the security procedure after reception of RAU REQ at Step 12 and transmission of RAU ACCEPT at Step 13. This needs to be implemented since the RAI of both cells is diferent, and the RAU REQ has come on a new rrc connection. A Prose CR will be raised for this.

6) The current TTCN implementation of RAU ACCEPT message at Step 13 sets the update result to “RA Updating”. It also does not include P-TMSI-1 which should be included as per the message content specified for Step 13. As a result of change in point 4 above, the update result should be set to “Combined RA/LA updated”. Also, P-TMSI-1 should be included in RAU ACCEPT.

	Summary of change
	1) Instead of checking for RAI in Step 4 against, v_RAI_1, checked for ?.
2) Instead of using template cs_GMM_Cause(tsc_GMM_Cause_28), used cs_GMM_Cause(tsc_GMM_Cause_28, '25'O)
3) Called functions f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch), f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA), f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell) and f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell) in this order. Please see screenshot. 
4) Updated the TTCN to expect the RAU REQ as a InitialDirectTransfer message, and also updated the expected update type to “Combined RA/LA updating with IMSI attach”. Please see screenshot.
5) Called the function f_UTRAN34_RRC_Security(utran34_CellB,false, ps_domain) before transmitting the RAU ACCEPT message at Step 14. Please see screenshot.
6) Instead of setting the IE update result to “RA Updating”, set it to “Combined RA/LA updated” in the RAU ACCEPT message. Also included P-TMSI-1 which is as per the Prose. Please see screenshot.

	Source of change
	UTRAN34_GMM_SIMTC.ttcn

	
	


Before:
	function fl_TC_12_4_2_3b_Body() runs on UTRAN_PTC
  {
    var template (value) MobileIdentity v_MobileIdPTMSI_1;
    var RRC_DATA_IND v_RRC_DataInd;
    var template (value) RRCDataReqType v_RAUAccept;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var PLMN_Identity v_PLMN_Identity_A;
    var PLMN_Identity v_PLMN_Identity_B;
    var octetstring v_PLMN_A;
    var octetstring v_PLMN_B;
    var octetstring v_LAC_A;
    var octetstring v_LAC_B;
    var octetstring v_RAC_A;
    var octetstring v_RAC_B;
    var template (present) RoutingAreaId v_RAI_1;
    var template (value) RoutingAreaId v_RAI_4;
    var PLMN_Identity v_EPLMN := {mcc := {0,0,2}, mnc := {0,1}};
    v_PLMN_Identity_A := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA);
    v_PLMN_A := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_A);
    v_LAC_A := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC_A := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_PLMN_Identity_B := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB);
    v_PLMN_B := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_B);
    v_LAC_B := f_UTRAN_CellInfo_GetLAC(utran34_CellB);
    v_RAC_B := f_UTRAN_CellInfo_GetRAC(utran34_CellB);
    v_RAI_1 := cr_RAI(omit, cr_LAI(omit, v_PLMN_A, v_LAC_A), v_RAC_A);
    v_RAI_4 := cs_RAI(omit, cs_LAI(omit, v_PLMN_B, v_LAC_B), v_RAC_B);
    v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def);
    // Set the cell type of cell A to the "Serving cell"
    // Set the cell type of cell B to the "Non-Suitable cell"
    // The UE is set in UE operation mode A (see ICS)
    // The UE is powered up or switched on and initiates an attach (see ICS). Cell A is preferred by the UE
    //@siclog "Step 1-3" siclog@
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);
    f_UTRAN34_SetUE_OperationMode(opModeA);
    //The UE is powered up or switched on and initiates an attach
    f_UT_SwitchOnUE(UT, false);
    // For Non auto attach UE: trigger the Attach procedure via an AT command.
    if (not(pc_AutomaticAttachSwitchON)) {
      f_Delay(5.0);
      f_UT_TriggerAttach(UT);
    }
    // RRC Connection with cause "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 4" siclog@
    //ATTACH REQUEST: Attach type = 'Combined GPRS/IMSI attach'
    //Mobile identity = IMSI;
    //TMSI status = no valid TMSI available
    //Additional update type  = SMS only
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeCombined),
                                                                                          f_Imsi2MobileIdentity(px_IMSI_Def),
                                                                                          v_RAI_1))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.attachRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.addUpdType)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Additional Update Type IE not present");
    }
    if (v_RRC_DataInd.msg.attachRequest.addUpdType.addUpdateTypeValue == '1'B){ // SMS only
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 12.4.2.3b, Step 4");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Additional Update Type IE is not set to SMS only");
    }
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 5-6" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 7" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 8-9 siclog@
    // ATTACH ACCEPT:
    // Attach result = GPRS only
    // Allocated P-TMSI = P-TMSI-2
    // P-TMSI Signature = P-TMSI-2 signature
    // Routing area identity = RAI-1
    // Additional network feature support = SMS via GPRS supported
    // GMM Cause = # 28 (SMS provided via GPRS in this routing area)
    f_UTRAN34_AttachAccept(utran34_CellA,
                           tsc_I_AttachTypeGPRSAtt,
                           tsc_PTMSI_2,
                           tsc_PTMSISig_2,
                           -,
                           -,
                           cs_GMM_Cause(tsc_GMM_Cause_28),
                           cs_AddNtwFeatureSupport('0'B));
    //@siclog "Steps 10-11" siclog@
    //Set the cell type of cell A to the "Suitable neighbour cell"
    //Set the cell type of cell B to the "Serving cell"
    //Cell B is preferred by the UE
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell);
    //Configure cell for RRC Connection establishment
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA);
    f_UTRAN34_RRC_ConnEst(utran34_CellB);
    //@siclog "Step 12" siclog@
    // ROUTING AREA UPDATE REQUEST
    // Update type = 'RA updating'
    // Old P-TMSI signature=P-TMSI-2 signature
    // Old Routing area identity = RAI-1
    // Additional update type  = SMS only
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U34_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeRAOnly), //@sic R5s150682 sic@
                                                                 v_RAI_1,
                                                                 cr_PTMSI_Signature(tsc_PTMSISig_2)))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.routingAreaUpdateRequest.addUpdType)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Additional update type IE not present");
    }
    if(v_RRC_DataInd.msg.routingAreaUpdateRequest.addUpdType.addUpdateTypeValue == '1'B){ // SMS Only
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 12.4.2.3b, Step 12");
    }
    else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Additional update type IE is not set to SMS Only");
    }
    //@siclog "Step 13" siclog@
    // ROUTING AREA UPDATE ACCEPT
    // Update result = RA updated
    // Mobile identity = P-TMSI-1
    // P-TMSI-1 signature
    // Routing area identity = RAI-4
    // Equivalent PLMNs = MCC2, MNC1
    // Additional network feature support = SMS via GPRS supported
    v_RAUAccept := cs_U_RA_UpdAcc3(cs_GMM_UpdateResultRA_Updated,
                                   v_RAI_4,
                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                   omit, //No new mobile identity
                                   omit);
    v_RAUAccept.routingAreaUpdateAccept.equivalentPLMN := cs_PLMN_List_1PLMN(f_U_Asn2Nas_PlmnId(v_EPLMN));
    v_RAUAccept.routingAreaUpdateAccept.addNtwFeatureSupport := cs_AddNtwFeatureSupport('0'B);
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             v_RAUAccept));
    //@siclog "Step 14" siclog@
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
  }


After:

	function fl_TC_12_4_2_3b_Body() runs on UTRAN_PTC
  {
    var template (value) MobileIdentity v_MobileIdPTMSI_1;
    var RRC_DATA_IND v_RRC_DataInd;
    var template (value) RRCDataReqType v_RAUAccept;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var PLMN_Identity v_PLMN_Identity_A;
    var PLMN_Identity v_PLMN_Identity_B;
    var octetstring v_PLMN_A;
    var octetstring v_PLMN_B;
    var octetstring v_LAC_A;
    var octetstring v_LAC_B;
    var octetstring v_RAC_A;
    var octetstring v_RAC_B;
    var template (present) RoutingAreaId v_RAI_1;
    var template (value) RoutingAreaId v_RAI_4;
    var PLMN_Identity v_EPLMN := {mcc := {0,0,2}, mnc := {0,1}};
    v_PLMN_Identity_A := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA);
    v_PLMN_A := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_A);
    v_LAC_A := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC_A := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_PLMN_Identity_B := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB);
    v_PLMN_B := f_U_Asn2Nas_PlmnId(v_PLMN_Identity_B);
    v_LAC_B := f_UTRAN_CellInfo_GetLAC(utran34_CellB);
    v_RAC_B := f_UTRAN_CellInfo_GetRAC(utran34_CellB);
    v_RAI_1 := cr_RAI(omit, cr_LAI(omit, v_PLMN_A, v_LAC_A), v_RAC_A);
    v_RAI_4 := cs_RAI(omit, cs_LAI(omit, v_PLMN_B, v_LAC_B), v_RAC_B);
    v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def);
    // Set the cell type of cell A to the "Serving cell"
    // Set the cell type of cell B to the "Non-Suitable cell"
    // The UE is set in UE operation mode A (see ICS)
    // The UE is powered up or switched on and initiates an attach (see ICS). Cell A is preferred by the UE
    //@siclog "Step 1-3" siclog@
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);
    f_UTRAN34_SetUE_OperationMode(opModeA);
    //The UE is powered up or switched on and initiates an attach
    f_UT_SwitchOnUE(UT, false);
    // For Non auto attach UE: trigger the Attach procedure via an AT command.
    if (not(pc_AutomaticAttachSwitchON)) {
      f_Delay(5.0);
      f_UT_TriggerAttach(UT);
    }
    // RRC Connection with cause "Registration"
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration);
    //@siclog "Step 4" siclog@
    //ATTACH REQUEST: Attach type = 'Combined GPRS/IMSI attach'
    //Mobile identity = IMSI;
    //TMSI status = no valid TMSI available
    //Additional update type  = SMS only
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeCombined),
                                                                                          f_Imsi2MobileIdentity(px_IMSI_Def),
                                                                                          ?/*WA#12_4_2_3b v_RAI_1*/))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.tmsiStatus)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE not present");
    }
    if (v_RRC_DataInd.msg.attachRequest.tmsiStatus.flag != tsc_NoValidTmsi){
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "TMSI Status IE is not set to 'No Valid TMSI'");
    }
    if (not ispresent(v_RRC_DataInd.msg.attachRequest.addUpdType)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Additional Update Type IE not present");
    }
    if (v_RRC_DataInd.msg.attachRequest.addUpdType.addUpdateTypeValue == '1'B){ // SMS only
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 12.4.2.3b, Step 4");
    } else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Additional Update Type IE is not set to SMS only");
    }
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 5-6" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 7" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Step 8-9 siclog@
    // ATTACH ACCEPT:
    // Attach result = GPRS only
    // Allocated P-TMSI = P-TMSI-2
    // P-TMSI Signature = P-TMSI-2 signature
    // Routing area identity = RAI-1
    // Additional network feature support = SMS via GPRS supported
    // GMM Cause = # 28 (SMS provided via GPRS in this routing area)
    f_UTRAN34_AttachAccept(utran34_CellA,
                           tsc_I_AttachTypeGPRSAtt,
                           tsc_PTMSI_2,
                           tsc_PTMSISig_2,
                           -,
                           -,
                           cs_GMM_Cause(tsc_GMM_Cause_28, '25'O/*WA#12_4_2_3b*/),
                           cs_AddNtwFeatureSupport('0'B));
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);//WA#12_4_2_3b
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA);//WA#12_4_2_3b
    //@siclog "Steps 10-11" siclog@
    //Set the cell type of cell A to the "Suitable neighbour cell"
    //Set the cell type of cell B to the "Serving cell"
    //Cell B is preferred by the UE
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell);
    //Configure cell for RRC Connection establishment
    //WA#12_4_2_3b f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA);
    f_UTRAN34_RRC_ConnEst(utran34_CellB);
    //@siclog "Step 12" siclog@
    // ROUTING AREA UPDATE REQUEST
    // Update type = 'RA updating'
    // Old P-TMSI signature=P-TMSI-2 signature
    // Old Routing area identity = RAI-1
    // Additional update type  = SMS only
    U_Dc.receive(//WA#12_4_2_3b car_PS_UplinkDirectTransfer
    


 car_PS_InitDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U34_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeCombinedWithIMSI 
                                             /*WA#12_4_2_3b tsc_I_UpdateTypeRAOnly*/), //@sic R5s150682 sic@
                                                                 v_RAI_1,
                                                                 cr_PTMSI_Signature(tsc_PTMSISig_2)))) -> value v_RRC_DataInd;
    if (not ispresent(v_RRC_DataInd.msg.routingAreaUpdateRequest.addUpdType)) {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Additional update type IE not present");
    }
    if(v_RRC_DataInd.msg.routingAreaUpdateRequest.addUpdType.addUpdateTypeValue == '1'B){ // SMS Only
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Test Case 12.4.2.3b, Step 12");
    }
    else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Additional update type IE is not set to SMS Only");
    }
    f_UTRAN34_RRC_Security(utran34_CellB,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           false,
                           ps_domain);//WA#12_4_2_3b
    //@siclog "Step 13" siclog@
    // ROUTING AREA UPDATE ACCEPT
    // Update result = RA updated
    // Mobile identity = P-TMSI-1
    // P-TMSI-1 signature
    // Routing area identity = RAI-4
    // Equivalent PLMNs = MCC2, MNC1
    // Additional network feature support = SMS via GPRS supported
    v_RAUAccept := cs_U_RA_UpdAcc3(cs_GMM_UpdateResultCombRA_LA, //WA#12_4_2_3b cs_GMM_UpdateResultRA_Updated,
                                   v_RAI_4,
                                   cs_PTMSI_Signature(px_PTMSI_SigDef),
                                   v_MobileIdPTMSI_1, //WA#12_4_2_3b omit, //No new mobile identity
                                   omit);
    v_RAUAccept.routingAreaUpdateAccept.equivalentPLMN := cs_PLMN_List_1PLMN(f_U_Asn2Nas_PlmnId(v_EPLMN));
    v_RAUAccept.routingAreaUpdateAccept.addNtwFeatureSupport := cs_AddNtwFeatureSupport('0'B);
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             v_RAUAccept));
    //@siclog "Step 14" siclog@
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));
  }


4.4 cs_GMM_Cause
	Record Name
	Cs_GMM_Cause

	Reason for change
	In the current TTCN implementation, at Step 8 the ATTACH ACCEPT message is sent with GMM cause “SMS provided via GPRS in this routing area” (#28). However, the template used for the GMM cause does not include the IEI. This needs to be changes since the GMM Cause is a TV type as per Table 9.4.2 in 24.008 sec 9.4.2. This needs to be changed

	Summary of change
	Parameterised the template cs_GMM_Cause so that it can take IEI as an argument. Set the default assignment for IEI to omit so that other usage of this template in other situations is unaffected.

	Source of change
	NAS_24008Templates_IRAT.ttcn

	
	


Before:
	template (value) GMM_Cause cs_GMM_Cause (B8_Type p_RejectCause) :=
  { // @sic R5s110718 sic@
  /* @status    APPROVED (LTE_IRAT, UTRAN) */
   iei := omit,
   gmmCauseValue := p_RejectCause
  };


After:

	template (value) GMM_Cause cs_GMM_Cause (B8_Type p_RejectCause,
  









   template (omit) IEI8_Type p_IEI := omit) :=
  { // @sic R5s110718 sic@
  /* @status    APPROVED (LTE_IRAT, UTRAN) */
   iei := p_IEI,
   gmmCauseValue := p_RejectCause
  };


5 Branches executed in test case 12.4.2.3b
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Intel 7360 device
The Intel 7360 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_12_4_2_3b_Intel_7360.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
7 References
	[1]
	R5s160725
This archive comprises Execution log files, PICS/PIXIT files


